
ORTOGRAFIA LANTZEN

Programa honen helburua zenbait hitzen orto-
grafia lantzea da. Aldi bakoitzean hitz bat azalduko
da pantailan. Hitz hori zuzen idatzita egon daiteke
edota akatsen bat eduki dezake. Eskatuko zaizuna bi
kasuetako edozeinetan hitza zuzen idaztea da. Pro-
gramak asmatu duzun ala ez adieraziko dizu, oker
zabiltzan kasuetan erantzun zuzena emanez.

Hemen aurkezten dizugun programa nahikoa
era sinplean egina dago eta hitz konkretu batzuk

landu dira. Gure asmo nagusia mota honetako pro-
gramak nola idatz ditzakezun azaltzea da; gero ba-
koitzak ikusiko du lantzea interesatzen zaizkion hitzak
zeintzuk diren eta aldaketaren bat egin nahi duen ala
ez.

Kasu honetan hitzak soilik landu badira ere,
badaude beste posibilitate batzuk: definizio bat
emanda definitutako hitza edo kontzeptua galdetu
edota testu bateko zenbait hitz zuzentzea eskatu.

10  REM ORTOGRAFIA LANTZEN
20      DIM FIL(5) : J=0
30      KEY OFF: SCREEN 1,0 : CLS
40      LOCATE 12,6 : PRINT "ORTOGRAFIA LANTZEKO PROGRAMA"
50      FOR I=1 TO 2000 : K=K+1 : NEXT I
60      CLS : RANDOMIZE TIMER
70      LOCATE 4,15 : PRINT "MENU NAGUSIA"
80      LOCATE 7,9  : PRINT "1. H-z edo H-rik gabe"
90      LOCATE 9,9  : PRINT "2. TX, TS edo TZ"
100     LOCATE 11,9 : PRINT "3. X, S edo Z"
110     LOCATE 13,9 : PRINT "4. A organikoa, bai ala ez?"
120     LOCATE 15,9 : PRINT "5. Guztiak nahasian"
130     LOCATE 17,9 : PRINT "6. Programa amaitu"
140     LOCATE 20,6 : INPUT "Zein aukera landu nahi duzu";AUK
150     IF AUK < 1 OR AUK > 6 THEN LOCATE 20,6 : PRINT SPC(34) : GOTO 140
160     CLS
170     LOCATE 2,12 : PRINT "ORTOGRAFIA LANTZEN"
180     LOCATE 7,6  : PRINT "Hona hemen aztertu behar duzun"
190     LOCATE 9,6  : PRINT "hitza:"
200     ONGI=0
210     REM BITARTEAN
220         IF (AUK<5 AND ONGI=15) OR (AUK=5 AND ONGI=50) THEN 470
230         REM AUKERA
240             IF AUK=1 THEN KOP=19 : RESTORE 540 : GOTO 300
250             IF AUK=2 THEN KOP=25 : RESTORE 560 : GOTO 300
260             IF AUK=3 THEN KOP=22 : RESTORE 590 : GOTO 300
270             IF AUK=4 THEN KOP=23 : RESTORE 610 : GOTO 300
280             IF AUK=5 THEN KOP=89 : RESTORE 540 : GOTO 300
290             GOTO 490
300         REM AMAUKERA
310         ALE = INT (RND*KOP) + 1
320         I=1 : AURKITUA$="FALTSUA"
330         WHILE AURKITUA$="FALTSUA" AND I<=J : IF FIL(I)=ALE THEN



AURKITUA$="EGIAZKOA" ELSE I=I+1 : WEND
340         IF AURKITUA$="EGIAZKOA" THEN 310
350         IF J < 5 THEN J=J+1 : FIL(J)=ALE : GOTO 370
360         FOR K=2 TO 5 : FIL(K-1)=FIL(K) : NEXT K
370         FIL(5) = ALE
380         FOR I=1 TO ALE : READ HITZA$,HITZA1$ : NEXT I
390         LOCATE 13,1 : PRINT SPC(39)
400         LOCATE 13, INT ((40-LEN(HITZA$))/2) : PRINT HITZA$
410         LOCATE 19,6 : PRINT "Idatz ezazu behar duen bezala:"
420         LOCATE 21,INT ((40-LEN(HITZA1$))/2) : INPUT "",ERAN$
430         LOCATE 23,6 : IF ERAN$=HITZA1$ THEN PRINT "ONGI!!!" : ONGI=ONGI+1

ELSE PRINT "GAIZKI! ";HITZA1$;" behar zuen."
440         FOR I=1 TO 2000 : K=K+1 : NEXT I
450         LOCATE 19,6 : PRINT SPC(31) : LOCATE 21,1 : PRINT SPC(39) :
              LOCATE 23,6 : PRINT SPC(31)
460         GOTO 210
470     REM AMBITARTEAN
480     GOTO 60
490     END
500 REM AMAIA ORTOGRAFIA LANTZEN
510                                                                       REM
520                                                                       REM
530 REM DATUAK
540     DATA
"neurri","neurri","ziur","ziur","mee","mehe","leun","leun","leoi","lehoi","leia","lehia","oiu","oihu","oian","oihan","oitura","ohitura","saiets","saihets","ilobi","hilobi","ildo","ildo","ile","ile","ila","hila","amar","hamar"
550     DATA "or","hor","urbil","hurbil","urrun","urrun","deiadar","deiadar"
560     DATA
"lotzagabe","lotsagabe","aintzakotzat","aintzakotzat","ahantzi","ahantzi","saihetz","saihets","itzusi","itsusi","hutzune","hutsune","saltza","saltsa","zintso","zintzo","atzeden","atseden","jaitzi","jaitsi","jauretzi","jauretsi"
570     DATA
"jauretze","jauretxe","goilaratzo","goilaratxo","igeltzero","igeltzero","zerbitzu","zerbitzu","itzali","itzali","laguntzaile","laguntzaile","pentzamendu","pentsamendu","errautz","errauts","fidagaitz","fidagaitz"
580     DATA "arratzalde","arratsalde","tzapel","txapel","erantzun","erantzun","erakutzi","erakutsi","tziminia","tximinia"
590     DATA
"bisi","bizi","serbitzu","zerbitzu","nahazketa","nahasketa","helebitazun","helebitasun","sazpi","zazpi","lazai","lasai","ahaztu","ahaztu","azkatu"
,"askatu","ezkondu","ezkondu","ezinezko","ezinezko","zitzaiskizuen","zitzaizkizuen"
600     DATA
"xakur","zakur","zikin","zikin","zabaldu","zabaldu","gezur","gezur","zei","sei","gaste","gazte","zahutu","xahutu","zukalde","sukalde","zagar","sagar","zarma","xarma","Gazteiz","Gasteiz"
610     DATA
"Donosti","Donostia","gauz","gauza","famili","familia","premi","premia","lengoai","lengoaia","energi","energia","zientzi","zientzia","kate","katea","arnas","arnasa","memori","memoria","makil","makila","makina","makina","makina","mail","maila"
620     DATA
"buru","buru","anai","anaia","bidai","bidaia","ibai","ibai","auzo","auzo","behi","behi","leun","leun","umil","umil","urri","urri","aztarna","aztarna"
630 REM AMAIA DATUAK

Oharrak:

1.- Programa honetan galderak egiteko garaian, hitzak alea-
torioki aukeratzen dira DATA sententziatan sartutako da-
tuetatik (310. lerroa).
Baina aleatoriotasunak badu eragozpen bat, eta hauxe da:
zenbakiak edozein momentutan errepika daitezke eta hori
gertatuz gero galdera berbera behin eta berriz errepika
daiteke segidan. Hau ebitatzeko iragazki edo filtro bat jarri
dugu; honekin azkeneko galdera aurrez egindako n galde-
ratatik desberdina izatea ziurtatzen da. Iragazkia taula batez
egin dugu. Aleatorio berri bat ateratzen den bakoitzean,
lehendik daudenekin konparatu beharko da, eta desberdina
bada taulan azken elementu bezala gorde, zaharrena (1.a)
kenduz. (320-370 lerroak).
Errepikaketak bakantzea nahi izanez gero, taularen luzera

handiagotzea nahikoa da. Kasu honetan 5 elementuko taula
erabili dugu.

2.- Kasu honetan erantzunak minuskulaz idatzi beharko dira
programak onar ditzan, baina maiuskulaz ere onartzeko ez
dago inongo eragozpenik. Hori lortzeko ondoko aldaketak
egin beharko dituzu:
–DATA sententziatan erantzuna maiuskulaz ere sartu.
–Irakurketa egiterakoan datuak hirunaka irakurri (380 le-
rroa)
–Erantzuna konparatzerakoan bi datuekin konparatu:
minuskulaz dagoenarekin eta maiuskulaz dagoenarekin.

3.- Programak eskaintzen dituen aukeretatik lehenengo 4.
aukerak amaitzeko baldintza 15 galdera zuzen erantzutea
da. 5. aukera amaitzeko berriz 50 asmatzea. (220 lerroa).
Aukera bat amaitzen den bakoitzean programa berrabiarazi
gabe beste batean sartzeko posibilitatea eskaintzen da (460
lerroa).


